Influence of age polyethism on longevity of workers in social insects
Appendix

Adam Tofilski

Bee Research Department
Agricultural University
29 Listopada 52
31-425 Krakow, Poland
rotofils@cyf-kr.edu.pl

This is a *Mathematica* 4.0 notebook.

1 Simplified model

It is assumed that there are two sets of tasks, A and B, which are associated with aging-independent mortality rates **mA** and **mB** respectively. A fixed proportion of time **f** is devoted to A-type tasks. A worker cannot spend more than the maximum resource **k** available for the whole life. The rates of resource expenditure during tasks A and B are **cA** and **cB** respectively. In the simplified model aging does not affect the mortality of workers until a certain age is reached, when resources become exhausted. At this time all workers die. In the model the expected longevity of workers in colonies with and without age polyethism is compared. If there is no polyethism the workers perform tasks A and B in turn. If age polyethism is present the workers perform A-type tasks first and after time **ts** they switch to B-type tasks.

1.1 General case

Error messages are switched off.

```
Off[General::spell1]
Off[General::unfl]
Off[Solve::ifun]
Off[NIntegrate::nlim]
```

The expected longevity of workers in colonies without polyethism is given by

$$p0 = \int_0^{\frac{k}{cA\,f + cB\,(1-f)}} \mathrm{Exp}[-(mA\,f + mB\,(1-f))\,t]\,dt$$

```
p0 = Simplify[%]
```

$$\frac{1}{f\,mA + mB - f\,mB} - \frac{e^{-\frac{k\,(f\,mA + mB - f\,mB)}{cB\,(1-f) + cA\,f}}}{f\,mA + mB - f\,mB}$$

$$-\frac{-1 + e^{\frac{k\,(f\,(mA - mB) + mB)}{cB\,(-1+f) - cA\,f}}}{f\,(mA - mB) + mB}$$

The expected longevity of workers during A-type tasks is given by

$$aA = \int_0^{ts} \mathrm{Exp}[-mA\,t]\,dt$$

```
aA = Simplify[%]
```

$$\frac{1}{mA} - \frac{e^{-mA\,ts}}{mA}$$

$$\frac{1 - e^{-mA\,ts}}{mA}$$

The expected longevity of workers during B-type tasks is given by

$$aB = \mathrm{Exp}[-mA\,ts] \int_0^{\frac{k - ts\,cA}{cB}} \mathrm{Exp}[-mB\,t]\,dt$$

```
aB = Simplify[%]
```

$$e^{-mA\,ts}\left(\frac{1}{mB}-\frac{e^{-\frac{mB\,(k-cA\,ts)}{cB}}}{mB}\right)$$

$$\frac{e^{-mA\,ts}\left(1-e^{-\frac{mB\,(k-cA\,ts)}{cB}}\right)}{mB}$$

The expected longevity of workers in colonies with age polyethis given by

```
pa = aA + aB;
```

The switching time `ts` cannot be found in a standard way.

$$\text{Solve}\left[\frac{aA}{aA+aB}==f,\ ts\right]$$

Solve::tdep : The equations appear to involve the variables to be solved for in an essentially non-algebraic way.

$$\text{Solve}\left[\frac{1-e^{-mA\,ts}}{mA\left(\frac{1-e^{-mA\,ts}}{mA}+\frac{e^{-mA\,ts}\left(1-e^{-\frac{mB\,(k-cA\,ts)}{cB}}\right)}{mB}\right)}==f,\ ts\right]$$

1.2 Solution

It was noticed that the expected longevity of workers in colonies with age polyethism is the same as in colonies without polyethism when

$$cB=\frac{mB\,cA}{mA};$$

The expected longevity of workers in colonies without polyethism is given by

$$p0s=\int_0^{\frac{k}{cA\,f+cB\,(1-f)}}\text{Exp}[-(mA\,f+mB\,(1-f))\,t]\,dt$$

$$p0s=\text{Simplify}[\%]$$

$$\frac{1}{f\,mA+mB-f\,mB}-\frac{e^{-\frac{k\,(f\,mA+mB-f\,mB)}{cA\,f+\frac{cA\,(1-f)\,mB}{mA}}}}{f\,mA+mB-f\,mB}$$

$$\frac{1-e^{-\frac{k\,mA}{cA}}}{f\,mA+mB-f\,mB}$$

The expected longevity of workers during A-type tasks is given by

$$aAs=\int_0^{ts}\text{Exp}[-mA\,t]\,dt$$

$$aAs=\text{Simplify}[\%]$$

$$\frac{1}{mA}-\frac{e^{-mA\,ts}}{mA}$$

$$\frac{1-e^{-mA\,ts}}{mA}$$

The expected longevity of workers during B-type tasks is given by

$$\mathbf{aBs} = e^{-mA\,ts} \int_0^{\frac{k - ts\,cA}{cB}} \mathbf{Exp[-mB\,t]}\, dt$$

aBs = Simplify[%]

$$e^{-mA\,ts}\left(\frac{1}{mB} - \frac{e^{-\frac{mA\,(k - cA\,ts)}{cA}}}{mB}\right)$$

$$-\frac{e^{-\frac{k\,mA}{cA}} - e^{-mA\,ts}}{mB}$$

When $\mathbf{cB} = \frac{\mathbf{mB\,cA}}{\mathbf{mA}}$ the switching time `ts` can be found analytically.

$$\mathbf{Solve}\left[\frac{\mathbf{aAs}}{\mathbf{aAs + aBs}} == \mathbf{f,\ ts}\right]$$

Simplify[%]

$$\left\{\left\{ts \to \frac{k\,mA + cA\,\mathrm{Log}\left[\frac{f\,mA + mB - f\,mB}{f\,mA + e^{\frac{k\,mA}{cA}}\,mB - e^{\frac{k\,mA}{cA}}\,f\,mB}\right]}{cA\,mA}\right\}\right\}$$

$$\left\{\left\{ts \to \frac{k}{cA} + \frac{\mathrm{Log}\left[\frac{f\,mA + mB - f\,mB}{f\,mA + e^{\frac{k\,mA}{cA}}\,mB - e^{\frac{k\,mA}{cA}}\,f\,mB}\right]}{mA}\right\}\right\}$$

It can be demonstrated that the expected longevity of workers in colonies with age polyethism is the same as in colonies without polyethism when $\mathbf{cB} = \frac{\mathbf{mB\,cA}}{\mathbf{mA}}$.

pas = aAs + aBs /. %[[1]]
pas = Simplify[%]

$$\frac{1 - e^{-mA\left(\frac{k}{cA} + \frac{\mathrm{Log}\left[\frac{f\,mA + mB - f\,mB}{f\,mA + e^{\frac{k\,mA}{cA}}\,mB - e^{\frac{k\,mA}{cA}}\,f\,mB}\right]}{mA}\right)}}{mA} - \frac{e^{-\frac{k\,mA}{cA}} - e^{-mA\left(\frac{k}{cA} + \frac{\mathrm{Log}\left[\frac{f\,mA + mB - f\,mB}{f\,mA + e^{\frac{k\,mA}{cA}}\,mB - e^{\frac{k\,mA}{cA}}\,f\,mB}\right]}{mA}\right)}}{mB}$$

$$\frac{1 - e^{-\frac{k\,mA}{cA}}}{f\,mA + mB - f\,mB}$$

p0s == pas

True

1.3 Numerical solution

The way in which the problem was solved does not imply that there is only one solution. Therefore the equations were solved numerically here.

Numerical values are assigned to parameters.

```
cA = 2;
cB = 4;
k = 200;
f = 0.5;
mA = 0.01;
```

Approximate function `tsi` represents the switching time `ts`.

$$\mathbf{tsv[(mB\_)?NumberQ] := ts\ /.\ FindRoot}\left[\frac{\mathbf{aA}}{\mathbf{aA + aB}} == \mathbf{f,\ \{ts,\ 5\}}\right]$$

```
tsi = FunctionInterpolation[tsv[mB], {mB, 0.001, 0.06}];
```

Switching time `ts` is replaced with approximate function `tsi`.

```
pa = aA + aB /. ts → tsi[mB];
```

The difference between expected longevity of workers in colonies with and without age polyethism is plotted.

```
Plot[pa - p0, {mB, 0.001, 0.06}, AxesLabel → {"mB", "pa-p0"}];
```



1.4 Absence of aging

If there is no aging (`cA` = `cB` = 0) the maximum life span is infinity.

The previous definition of the parameters have to be removed to allow analytical computation.

```
cB =.
cA =.
cB =.
k =.
f =.
mA =.
```

The expected longevity of workers in colonies without polyethism is given by

```
p00 = ∫_0^∞ Exp[-(mA f + mB (1 - f)) t] ⅆt
p00 = Simplify[% /. {Re[f mA + mB - f mB] > 0 → True}]
```

$$\text{If}\left[\text{Re}[f\,mA + mB - f\,mB] > 0,\ \frac{1}{f\,mA + mB - f\,mB},\ \int_0^\infty e^{(-f\,mA-(1-f)\,mB)\,t}\,dt\right]$$

$$\frac{1}{f\,mA + mB - f\,mB}$$

The expected longevity of workers during A-type tasks is given by

```
aA0 = ∫_0^ts Exp[-mA t] ⅆt
aA0 = Simplify[%]
```

$$\frac{1}{mA} - \frac{e^{-mA\,ts}}{mA}$$

$$\frac{1 - e^{-mA\,ts}}{mA}$$

The expected longevity of workers during B-type tasks is given by

```
aB0 = Exp[-mA ts] ∫_0^∞ Exp[-mB t] ⅆt
aB0 = % /. {Re[mB] > 0 → True}
```

$$e^{-mA\,ts}\ \text{If}\left[\text{Re}[mB] > 0,\ \frac{1}{mB},\ \int_0^\infty e^{-mB\,t}\,dt\right]$$

$$\frac{e^{-mA\,ts}}{mB}$$

When there is no aging the switching time ts can be found analytically.

```
Solve[aA0/(aA0 + aB0) == f, ts]
Simplify[%]
```

$$\left\{\left\{ts \to \frac{\text{Log}\left[\frac{-f\,mA - mB + f\,mB}{(-1+f)\,mB}\right]}{mA}\right\}\right\}$$

$$\left\{\left\{ts \to \frac{\text{Log}\left[\frac{f\,mA + mB - f\,mB}{mB - f\,mB}\right]}{mA}\right\}\right\}$$

When there is no aging the expected longevity of workers in colonies with age polyethism is the same as in colonies without polyethism.

```
pa0 = aA0 + aB0 /. %[[1]]
pa0 = Simplify[%]
```

$$\frac{mB - f\,mB}{mB\,(f\,mA + mB - f\,mB)} + \frac{1 - \frac{mB - f\,mB}{f\,mA + mB - f\,mB}}{mA}$$

$$\frac{1}{f\,mA + mB - f\,mB}$$

```
p00 == pa0
```

True

2 More realistic model

It is assumed that mortality rate r increases exponentially with age t: $r = m + \alpha t^{\beta}$ where m is aging-independent mortality. Aging-related mortality is described by two parameters: α controlling the magnitude of this mortality, and β controlling the shape of the curve depicting changes of aging-related mortality with age. α associated with tasks A and B is αA and αB respectively. Tasks A and B are associated with the same β.

The expected longevity of workers during A-type tasks is given by

$$aAx = \int_0^{ts} \text{Exp}\left[-mA\,t - \frac{\alpha A\,t^{\beta+1}}{\beta+1}\right] dt;$$

The expected longevity of workers during B-type tasks is given by

$$aBx = \int_0^{\infty} \text{Exp}\left[-mB\,(t + tm) - \frac{\alpha B\,(t + t\alpha)^{\beta+1}}{\beta+1}\right] dt;$$

The correction terms tm and tα adjust the survivorship at the beginning of B-type tasks period to that the end of A-type tasks period.

```
Solve[Exp[-mB tm] == Exp[-mA ts], tm]
tm = tm /. %[[1]] /. Log[Exp[-mA ts]] → -mA ts
```

$$\left\{\left\{tm \to -\frac{\text{Log}\left[e^{-mA\,ts}\right]}{mB}\right\}\right\}$$

$$\frac{mA\,ts}{mB}$$

$$\text{Solve}\left[\text{Exp}\left[-\frac{\alpha B\,t\alpha^{\beta+1}}{\beta+1}\right] == \text{Exp}\left[-\frac{\alpha A\,ts^{\beta+1}}{\beta+1}\right],\ t\alpha\right]$$

$$t\alpha = t\alpha\ /.\ \%[[1]]\ /.\ \text{Log}\left[\text{Exp}\left[-\frac{ts^{1+\beta}\,\alpha A}{1+\beta}\right]\right] \to -\frac{ts^{1+\beta}\,\alpha A}{1+\beta}$$

$$\left\{\left\{t\alpha \to \left(-\frac{(1+\beta)\,\mathrm{Log}\left[e^{-\frac{ts^{1+\beta}\,\alpha A}{1+\beta}}\right]}{\alpha B}\right)^{\frac{1}{1+\beta}}\right\}\right\}$$

$$\left(\frac{ts^{1+\beta}\,\alpha A}{\alpha B}\right)^{\frac{1}{1+\beta}}$$

The expected longevity of workers in colonies without polyethism `P0` depends on probability of survival `sm` associated with aging-independent mortality and probability of survival `sα` associated with aging-related mortality. The probability of survival to age `t` associated with aging-independent mortality is given by

```
sm = Exp[-(mA f + mB (1 - f)) t];
```

The probability of survival to age `t` associated with aging-related mortality equals the probability of surviving B-type tasks during proportion (1-`f`) of time `t`, taking into account the correction term `tα`, where `ts` is replaced by the proportion `f` of time `t` spent performing the A-type tasks.

$$s\alpha = \mathrm{Exp}\left[-\frac{1}{\beta+1}\,\alpha B\left((1-f)\,t + \sqrt[\beta+1]{\frac{\alpha A}{\alpha B}}\;f\,t\right)^{\beta+1}\right];$$

The expected longevity of workers in colonies without polyethism is given by

$$p0x = \int_0^\infty sm\; s\alpha \, dt;$$

Numerical values were assigned to the parameters.

```
αA = 1.*^-7;
αB = 1.*^-6;
β = 3;
f = 0.5;
mA = 0.01;
```

Approximate function `tsi` represents the switching time `ts`.

$$tsv[(mB\_)?NumberQ] := ts\ /.\ \mathrm{FindRoot}\left[\frac{aAx}{aAx + aBx} == f,\ \{ts,\ 20\}\right]$$

```
tsi = FunctionInterpolation[tsv[mB], {mB, 0.001, 0.06}];
```

To simplify computations the expected longevity of workers in colonies with age polyethism is expressed as

$$pax = \frac{aAx}{f};$$

Switching time `ts` is replaced with approximate function `tsi`.

```
pax = pax /. ts → tsi[mB];
```

The difference in expected longevity of workers between colonies with and without age polyethism is plotted.

```
Plot[pax-p0x,{mB, 0.001, 0.06},AxesLabel->{"mB","pax-p0x"}];
```